```
BeginPackage["deltasimplify`"]

(*
  deltasimplify.m
    This package defines objects and procedures to carry out spin sums.

    Programmer:  Dick Furnstahl furnstahl.1@osu.edu

    Revision history:
      09-Feb-2003  --- original version
      03-Nov-2009  --- updated with g --> nu

*)

Unprotect["deltasimplify`*"]
Clear["deltasimplify`*"]

(*************************************************************************)

DeltaSimplify::usage = "DeltaSimplify[e] simplifies a spin sum"

del::usage = "del[a,b] is a Kronecker delta function with indices a,b."

(*************************************************************************)

Begin["`Private`"]

deltasimprules = {
  x_ (del[a_,b_]+y_) :> x*del[a,b] + x*y,
  del[a_,b_]*del[b_,c_] :> del[a,c],
  del[a_,b_]*del[c_,b_] :> del[a,c],
  del[a_,b_]*del[a_,c_] :> del[b,c],
  del[a_,b_]*del[c_,a_] :> del[b,c],
  del[a_,b_]*del[a_,b_] :> del[a,a],
  del[a_,b_]*del[b_,a_] :> del[a,a],
  del[a_,a_] :> -Global`nu
}

DeltaSimplify[e_] := FixedPoint[ Expand[# //. deltasimprules]&, e]

End[(* "`Private`" *)]

(*************************************************************************)

Protect["deltasimplify`*"]

EndPackage[]
```

anything

$\square(\delta_{ab} + \triangle) \to \square * \delta_{ab} + \square * \triangle$

$\delta_{ab}\delta_{ca} \to \delta_{bc}$

$\delta_{ab}\delta_{ba} \to \delta_{aa}$

$\delta_{aa} \to -\nu$

apply rules to expression e until it stops changing.

Doing Spin Sums I

In this notebook, we evaluate the spin sums for the lowest-order diagrams for a spin-independent delta function potential.

Programmer: Dick Furnstahl furnstahl.1@osu.edu Revision history: 03-Nov-2009 --- new version for 880.05

■ Load simplification package

The package is assumed to be in the same directory as this notebook. Load it and check definitions:

```
<< deltasimplify.m
```

```
? del
```

---

del[a,b] is a Kronecker delta function with indices a,b.

```
? DeltaSimplify
```

DeltaSimplify[e] simplifies a spin sum

■ Example Spin Sums



This first one is the "bowtie" diagram (leading order).

```
spinsum1 = DeltaSimplify[del[a, g] del[b, d] (del[a, g] del[b, d] + del[a, d] del[b, g])]
```

$-\nu + \nu^2$

From g^2's

$a \to \alpha$
$b \to \beta$
$d \to \delta$
$g \to \gamma$

```
Factor[spinsum1]
```

```
(-1 + nu) nu
```

This next one is the 2nd order bubble chain (anomalous).

```
spinsum2 = Factor[DeltaSimplify[del[a1, b1] del[a2, b3]
    del[a3, b2] del[a4, b4] (del[a1, b1] del[a2, b2] + del[a1, b2] del[a2, b1])
    (del[a3, b3] del[a4, b4] + del[a3, b4] del[a4, b3])]]
```



$-(-1 + \nu)^2 \nu$

This one is the "beachball" diagram.

```
spinsum3 = Factor[DeltaSimplify[del[a1, b1] del[a2, b2]
    del[a3, b3] del[a4, b4] (del[a1, b3] del[a2, b4] + del[a1, b4] del[a2, b3])
    (del[a3, b1] del[a4, b2] + del[a3, b2] del[a4, b1])]]
```



```
2 (-1 + nu) nu
```

This one is the Hartree-Fock diagram with a 3NF vertex.

```
spinsum4 = Factor[DeltaSimplify[del[a1, b1] del[a2, b2] del[a3, b3]
    (del[a1, b1] del[a2, b2] del[a3, b3] + del[a1, b1] del[a2, b3] del[a3, b2] +
      del[a1, b2] del[a2, b1] del[a3, b3] + del[a1, b3] del[a2, b2] del[a3, b1] +
      del[a1, b3] del[a2, b1] del[a3, b2] + del[a1, b2] del[a2, b3] del[a3, b1])]]
```

$-(-2 + \nu)(-1 + \nu)\nu$